# $Id: PKGBUILD 78640 2012-10-22 12:45:21Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Thomas Dziedzic < gostrc at gmail >
# Contributor: Sergio A. Morales <sergiomorales@archlinux.cl>
# Contributor: soeren <nonick@posteo.de>

pkgname=gummi
pkgver=0.6.5
pkgrel=1
pkgdesc='Simple LaTex editor for GTK users'
arch=('x86_64' 'i686')
url='http://gummi.midnightcoding.org/'
license=('MIT')
depends=('texlive-core' 'gtkspell' 'gtksourceview2' 'poppler-glib' 'desktop-file-utils')
makedepends=('glib2' 'gtk2' 'pango' 'poppler' 'intltool')
install=$pkgname.install
source=("$pkgname-$pkgver.tar.gz::http://dev.midnightcoding.org/projects/$pkgname/latest.tar.gz")
sha256sums=('b23c2958376ea43c701a276ad19ceac5b50d9cb32a489a10897b25aa5004fffb')
 
build() {
  cd $pkgname-$pkgver

  find . -type f -exec touch {} \;
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver

  make prefix=$pkgdir/usr install
  install -Dm644 COPYING $pkgdir/usr/share/licenses/$pkgname/COPYING
}

# vim:set ts=2 sw=2 et:
